

Polynome mit ganzzahligen Null-, Extrem- und Wendestellen

J. Meyer
(Hameln)

j.m.meyer@t-online.de



Einleitung

Mitunter sucht man Polynome f , deren besondere Stellen ganzzahlig sind, d.h. bei denen f , f' und f'' jeweils ganzzahlige Nullstellen haben. Solche Polynome sollen hier „ganzheitlich“ genannt werden. Für den Grad 2 ist deren Auffinden völlig problemlos, Grad 3 lässt sich mit CAS-Hilfe schnell klären, und bei Grad 4 versagt ein (naiv aufgesetztes) CAS. Der Einsatz eines CAS verführt dazu, relativ theoriearm mit Hilfe von „brute force and ignorance“ die Maschine einfach rechnen zu lassen. Allerdings können die dadurch erzielten Resultate durchaus nachträglich zu Überlegungen führen, ob der große Rechneraufwand tatsächlich notwendig gewesen wäre. Dies wird anhand von Grad 3 erläutert.

Insofern will dieser Artikel einen Beitrag leisten zur Diskussion, ob im Schulunterricht der Einsatz von CAS mathematische Überlegungen sogar verhindern kann. Andererseits lässt sich dieser Artikel natürlich auch auffassen als Hilfe für Lehrende, die ganzheitliche Polynome suchen. Selbstverständlich kann man den Inhalt dieses Beitrages auch mit Lernenden, die das frei erhältliche CAS „Maxima“ lernen wollen, erarbeiten.

Kubische Polynome

Da die Nullstellen der gesuchten kubischen Polynome ganzzahlig sein sollen, bietet es sich an, diese in faktorisierter Form anzusetzen. Man erleichtert dem Rechner die spätere Sucharbeit, wenn 0 eine der Nullstellen ist, denn wenn man Polynome entlang der Argumentachse ganzzahlig verschiebt, ändert sich nichts an deren Ganzheitlichkeit. Sind u und v die beiden anderen Nullstellen, setzt man

$$f(x) = x \cdot (x - u) \cdot (x - v)$$

an. Ein anderer höchster Koeffizient als 1 würde an der Ganzheitlichkeit nichts ändern. Da zusammenfallende Nullstellen das Problem stark vereinfachen würden, soll

vorausgesetzt werden. Nun kann man von einem CAS überprüfen lassen, für welche Wahlen von u und v die Ableitungen f' und f'' ganzzahlige Nullstellen haben. Das mag bei kubischen Polynomen f noch gut leistbar sein; bei Polynomen vom Grad 4 hat man es dann mit kubischen Gleichungen zu tun, und das sieht nach Programmieraufwand aus. Aber warum Gleichungen lösen? Stattdessen kann man das CAS auch alle Werte zwischen 0 und v überprüfen lassen, ob sie Nullstellen der beiden Ableitungen von f sind (Abb. 1).

```
n:50$
for u:1 thru n do
for v:u+1 thru n do
block
(f(x):=x*(x-u)*(x-v),
define(f1(x), diff(f(x), x)),
define(f2(x), diff(f1(x), x)),
Est:makelist(),
WSt:makelist(),
for x:1 thru n do
if f1(x)=0 then Est:endcons(x, Est),
if length(Est)=2
then for x:Est[1] thru Est[2] do
if f2(x)=0 then WSt:endcons(x, WSt),
if length(Est)*length(WSt)=2
then print(u, v, Est, WSt));
9 24 [ 4, 18 ] [ 11 ]
15 24 [ 6, 20 ] [ 13 ]
18 48 [ 8, 36 ] [ 22 ]
21 45 [ 9, 35 ] [ 22 ]
24 45 [ 10, 36 ] [ 23 ]
30 48 [ 12, 40 ] [ 26 ]
```

Abbildung 1

Die vorletzte Eingabezeile verwendet die Einsicht, dass die Wendestelle zwischen den beiden Extremstellen liegen muss. Man braucht hier nicht von Extrem- und Wendestellen-Kandidaten reden, da jede kubische Funktion eine Wende- und höchstens zwei Extremstellen hat; hat man diese Stellen gefunden, sind die hinreichenden Kriterien automatisch erfüllt. Betrachtet man die Ergebnisse, fällt eine Symmetrie auf: Ist

$$0 < u < v$$

$$(0, u, v)$$

ein geeignetes Tripel, so offenbar auch

$$(0, v - u, v),$$

und die Begründung ist einfach: Die Wendestelle ist durch

$$x_w = \frac{u + v}{3}$$

gegeben. Wegen

$$(u + v) + (v - u + v) = 3 \cdot v$$

ist $u + v$ genau dann durch 3 teilbar, wenn auch $v - u + v$ es ist. Die Extremstellen sind durch

$$x_e = x_w \pm \sqrt{x_w^2 - \frac{u \cdot v}{3}} = x_w \pm \frac{\sqrt{u^2 - u \cdot v + v^2}}{3}$$

gegeben. Wegen

$$(v - u)^2 - (v - u) \cdot v + v^2 = v^2 - u \cdot v + v^2$$

wird die Ganzheitlichkeit durch die Ersetzung von $(0, u, v)$ durch $(0, v - u, v)$ keineswegs beeinträchtigt.

Nun kann man sich weiter überlegen, ob der Aufwand von Abb. 1 notwendig war. Die Wendestelle x_w ist ganzzahlig, wenn $u = 3 \cdot a$ und $v = 3 \cdot b$ oder wenn $u = 3 \cdot a + 1$ und $v = 3 \cdot b + 2$ (oder umgekehrt) ist. Im zweiten Fall ist aber

$$\frac{u \cdot v}{3} = \frac{9 \cdot a \cdot b + 6 \cdot a + 3 \cdot b + 2}{3}$$

nicht ganzzahlig.

Somit sollte $u = 3 \cdot a$ und $v = 3 \cdot b$ sein. Dann gilt

$$x_w = a + b,$$

$$x_e = a + b \pm \sqrt{a^2 - a \cdot b + b^2};$$

es muss also

$$a^2 - a \cdot b + b^2$$

ein Quadrat sein. Hier hilft wieder ein systematisches Durchsuchen (Abb. 2).

```
n:15$
L:makelist();
for a:1 thru n do
for b:a+1 thru n do
block
(u:a^2-a*b+b^2,
if (floor(sqrt(u)+0.5))^2=u
then L:append(L,makelist([a,b]))$
L;
[]
[ [3, 8], [5, 8], [7, 15], [8, 15] ]
```

Abbildung 2

Das erste Ergebnis in Abb. 2 ($a = 3$ und $b = 8$) entspricht dem ersten Ergebnis von Abb. 1 ($u = 9$ und $v = 24$). Natürlich kann man in Abb. 2 eine deutlich höhere Grenze als $n = 15$ wählen.

Quartische Polynome

Man könnte eine zu Abb. 1 analoge Suchstruktur aufstellen (Abb. 3).

```
n:500$
for u:1 thru n do
for v:u+1 thru n do
for w:v+1 thru n do
block
(f(x):=x*(x-u)*(x-v)*(x-w),
define(f1(x), diff(f(x), x)),
define(f2(x), diff(f1(x), x)),
Est:makelist(),
WSt:makelist(),
for x:u+1 thru n do
if f1(x)=0 then Est:endcons(x, Est),
if length(Est)=3 then
for x:Est[1] thru Est[3] do
if f2(x)=0 then WSt:endcons(x, WSt),
if length(Est)*length(WSt)=6
then print(u, v, w, Est, WSt));
```

Abbildung 3

Leider stellt man fest, dass das CAS keine Lösungen findet. Es scheint auch so zu sein, dass bisher niemand weiß, ob es ganzheitliche quartische Polynome mit unterschiedlichen Nullstellen überhaupt gibt, vgl. etwa [1], die dort angegebene Literatur.

Da bleibt nur, mehrfache Nullstellen zuzulassen; aus Gründen der Einfachheit sei 0 die mehrfache Nullstelle. Es sind mehrere Fälle zu unterscheiden:

1. Eine doppelte und zwei einfache Nullstellen

Hier muss man lange suchen; der CAS-Code in Abb. 4 dürfte selbsterklärend sein.

```
n:500$
for u:1 thru n do
for v:u+1 thru n do
block
(f(x):=x*x*(x-u)*(x-v),
define(f1(x), diff(f(x), x)),
define(f2(x), diff(f1(x), x)),
Est:makelist(),
WSt:makelist(),
for x:0 thru n do
if f1(x)=0 then Est:endcons(x, Est),
if length(Est)=3 then
for x:Est[1] thru Est[3] do
if f2(x)=0 then WSt:endcons(x, WSt),
if length(Est)*length(WSt)=6
then print(u, v, Est, WSt));
```

Abbildung 4

Das kleinste Ergebnis ist $u = 308$, $v = 360$ mit Extremstellen bei 0, 165 und 336 sowie Wendestellen bei 70 und 264. Das Suchverfahren lässt sich beschleunigen: Ist

$$f(x) = x^2 \cdot (x - u) \cdot (x - v),$$

so hat f''' die Lösung

$$x_h = \frac{u + v}{4}$$

und f'' die Lösungen

$$x_w = x_h \pm \sqrt{x_h^2 - \frac{u \cdot v}{6}}$$

sowie f' die nichttrivialen Lösungen

$$x_e = \frac{3}{2} \cdot x_h \pm \sqrt{\left(\frac{3}{2} \cdot x_h\right)^2 - \frac{u \cdot v}{2}}.$$

Man bekommt Beispiele zur Ganzheitlichkeit, wenn x_h durch 2 teilbar ist (aber, wie das Beispiel $u = 308$ und $v = 360$ zeigt, muss das nicht sein!) und $u \cdot v$ durch 6; ferner müssen die Radikanden Quadrate sein (Abb. 5).

```
teilt(a,b):=if floor(b/a+0.5)=b/a
then true else false$

istQua(a):=if (floor(sqrt(a)+0.5))^2=a
then true else false$

n:1000$
L:makelist();
for u:500 thru n do
for v:u+1 thru n do
block
(h:(u+v)/4,
d:3*h/2,
b1:teilt(2, h),
b2:teilt(6, u*v),
if (b1 and b2) then
block
(r1:h^2-u*v/6,
b3:istQua(r1),
r2:d^2-u*v/2,
b4:istQua(r2),
bb:(b3 and b4),
if bb then
L:append(L,makelist([u, v])))$
L;
[ [616, 720] ]
```

Abbildung 5

Wenn man einschränkende Bedingungen macht, wird das Suchverfahren noch schneller: Es sei $u = 8 \cdot a$ und $v = 24 \cdot b$; die Radikanden sind

$$x_h^2 - \frac{u \cdot v}{6} = 4 \cdot ((a + 3 \cdot b)^2 - 8 \cdot a \cdot b)$$

und

$$\left(\frac{3}{2} \cdot x_h\right)^2 - \frac{u \cdot v}{2} = 9 \cdot (a + 3 \cdot b)^2 - 96 \cdot a \cdot b.$$

Abb. 6 zeigt zwei Ergebnisse unter $n = 100$; das erste entspricht der dem Ergebnis von Abb. 5.

```
n:200$
L:makelist();
for a:1 thru n do
for b:1 thru n do
block
```

```
(u:8*a, v:24*b, xh:(u+v)/4, d:3*xh/2,
r1:xh^2-u*v/6,
r2:d^2-u*v/2,
b1:(floor(sqrt(r1)+0.5))^2=r1,
b2:(floor(sqrt(r2)+0.5))^2=r2,
bb:b1 and b2,
if bb then
L:append(L, makelist([a, b]))$
L;
[ [77, 30], [154, 60] ]
```

Abbildung 6

2. Zwei doppelte Nullstellen

Man sieht an Abb. 7, dass es hier keine ganzheitlichen Beispiele geben kann, denn f'' hat in keinem Fall ganzzahlige Nullstellen.

```
f:x^2*(x-6*u)^2$
f1:diff(f,x)$
f2:diff(f1,x)$

solve(f2,x);
[x=(3-√3)u, x=(√3+3)u ]

solve(f1,x);
[x=3u, x=6u, x=0 ]
```

Abbildung 7

3. Eine dreifache Nullstelle

Dies ist ganz einfach: Abb. 8 zeigt, dass die einfache Nullstelle durch 4 teilbar sein muss.

```
f:x^3*(x-4*u)$
f1:diff(f,x)$
f2:diff(f1,x)$

solve(f2,x);
[ x=2u, x=0 ]

solve(f1,x);
[ x=3u, x=0 ]
```

Abbildung 8

Schlussbemerkung

Schon die Suche nach ganzheitlichen Polynomen vom Grad 4 erwies sich als unerwartet schwierig, so dass man sich nicht trauen wird, ganzheitliche Polynome vom Grad 5 ohne jeglichen Theorieaufwand zu suchen.

Literaturverzeichnis

- [1] Ralph Buchholz und James MacDougall. When Newton met Diophantus : A study of rational-derived polynomials and their extension to quadratic fields. *Journal of Number Theory* 81 (2000), no. 2, pp. 210-233.